Abstract

Unit testing has become a require part of software development. Every method that might possibly go wrong should have one or more unit tests that describe its behaviour. These tests act as executable documentation: they describe what the method should do in readable terms, by giving examples. If the tests pass, we can be sure that the implementation conforms to this specification. Of course, specification by example cannot be complete, and testing cannot ensure correctness, but it helps enormously in finding bugs, and speeds up development.

The unit testing framework for Grace is called GUnit. This document outlines briefly how to use it.

1 An Example

The best way to explain how to use GUnit is probably by example:

```grace
import "GUnit" as GU
import "set_vector" as SV

class aSetTest.forMethod(m) {
  inherits GU.aTestCase.forMethod(m)

  var emptySet // :Set<Number>
  var set23 // :Set<Number>

  method setup {
    super.setup
    emptySet := SV.SetVector.new
    set23 := SV.SetVector.new
    set23.add(2)
    set23.add(3)
  }
}  
```

method testEmpty {
    assert (emptySet.size == 0) description ("emptySet is not empty!")
    deny (emptySet.contains(2)) description ("emptySet contains 2!")
}

method testNonEmpty {
    assert (set23.size) shouldBe (2)
    assert (set23.contains(2))
    assert (set23.contains(3))
}

method testDuplication {
    set23.add(2)
    assert (set23.size == 2) description "duplication of 2 not detected by set"
}

GU.aTestSuite.fromTestMethodsInClass(aSetTest).runAndPrintResults

Line 1 imports the GUnit package, so that we can use it to implement our tests. Line 2 imports the package under test, here an implementation of Sets called set_vector.

Starting on line 4 we define a class whose instances will be the individual tests. The actual tests are methods in this class; the names of these methods are chosen to describe what is being tested. We call a class that contains test methods a test class.

After the definition of the class, the last line of the example runs the tests. It’s a bit complicated, so let’s look at it in parts.

A TestSuite (pronounced like “sweet”) is a collection of tests. Like an individual test, you can run a test suite, which runs all of the tests that it contains. There are several ways of making a TestSuite; here we use the method fromTestMethodsInClass on the object aTestSuite. This method takes as its argument the test class; it builds a test suite containing one test for each of the test methods in the test class.

What do we do with the test suite once we’ve made it? Run all of the tests, and print the results! There are other things that we could do too; we will look at test suites later in more detail.

2 Test Classes and Test Methods

Why are our tests methods of a class, rather than just objects with a run method? It’s sometimes useful to run a test more than once (for example, when hunting for a bug), and its impotent that each test should start with a “clean slate”, and not be contaminated by previous failed tests. So GUnit needs to be able to generate a new instance of a test when required. This is the function of the test class, here the class called aSetTest
What makes a class a **test class**?

1. Its instances inherit from `aTestCase.forMethod(m)`.
2. Its constructor method is called `forMethod(m)`.
3. Its instances have **methods** corresponding to the tests that we want to run; these **test methods** have no parameters and **must** have a name starting with **test**. For example, starting on line 18 we see a test called `testEmpty`.
4. It’s conventional to name the class after the class or object that it’s testing, and to include the word **Test** as a suffix. In our example, the class is called `aSetTest` because it’s testing the class `aSet`.
5. The test class can have method **setup** and **teardown**; if they exist, these methods will be run before and after each test method (whether or not the test passes).
6. The test class can have fields and other methods as necessary. For example, its sometimes convenient to have helper methods for clarity, such as `asset()isApproximatelyEqualTo()`.

What’s in a **test method**?  The only thing that has to be in a test is one or more **assertions**, which are self-requests of various assertion methods inherited from `aTestCase.forMethod(m)`.

```plaintext
method assert (bb: Boolean) description (message)
// asserts that bb is true. If bb is not true, the test will fail with message
method deny (bb: Boolean) description (message)
// asserts that bb is false. If bb is not false, the test will fail with message
method assert (bb: Boolean)
method deny (bb: Boolean)
// short forms, with the default message "assertion failure"
method assert (s1:Object) shouldBe (s2:Object)
// like assert (s1 == s2), but with a more appropriate default message
method assert(block0)shouldRaise(desiredException)
// asserts that the desiredException is raised during the execution of block0
method assert (block0) shouldntRaise (undesiredException)
// asserts that the undesiredException is not raised during the execution of block0.
// The assertion holds if block0 raises some other exception, or if it completes
// execution without raising any exception.
method failBecause (message)
// equivalent to assert (false) description (message)
method failure
// the exception object raised by a failed assertion
```

In addition to the assertions, a test can contain arbitrary executable code. However, because part of the function of a test is to serve as documentation, its a good idea to keep tests as simple as possible.

What happens when a **test runs**?  In general, one of three things might happen when a test runs.
1. The test passes, that is, all of the assertions that it makes are true.

2. The test fails, that is, one of the assertions is false.

3. The test errors\(^1\), that is, a runtime error occurs that prevents the test from completing. For example, the test may request a method that does not exist in the receiver, or might index an array out of bounds.

In all cases, GUnit will record the outcome, and then go on to run the next test. This is important, because we generally want to be able to run a suite of tests, and see how many pass, rather than have testing stop on the first error or failure. For example, when we run the set test suite shown above, we get the output

3 run, 0 failed, 0 errors

**What happens when your tests don’t pass?**  Suppose that we add another test to aSetTest:

```java
ten method testRemove {
    set23.remove(2)
    deny (set23.contains(3)) description "{set23} contains 3 after it was removed"
}
```

When we run the tests again, we get this output, which summarizes the test run:

```
4 run, 0 failed, 1 error
Errors:
    testRemove
```

The summary output will contain a list of all of the tests that failed, and a list of all of the tests that errored, but not a lot of debugging information. To get more information on the tests that don’t pass, we debug them. To do this, we add the following line of code to the test module:

```java
    aSetTest.forMethod("testRemove").debugAndPrintResults
```

This creates a test suite that contains a single method (the method testRemove, named in the string argument to forMethod), and debugs it. Here is the output:

```
4 run, 0 failed, 1 error
Errors:
    testRemove

debbuging method testRemove ...
Error around line 37: RuntimeError: Method lookup error: no remove in SetVector.
    Called aSetTest.debugAndPrintResults (defined at GUnit:150 in object at SetTests:44) on line 156
    Called aSetTest.debug (defined at GUnit:133 in object at SetTests:44) on line 152
    Called Block«GUnit:135».apply (defined at unknown:0 in object at unknown:0) on line 141
    Called Block«GUnit:135»._apply (defined at GUnit:129 in object at unknown:0) on line 141
```

\(^1\)Yes, I’m using “to error” as a verb. How daring!
When we *debug* a test, we see the error message, but any code subsequent to the test that didn’t pass is not run.

In this example, we see that the problem is that the object under test doesn’t actually have a method called *remove*. If we implement this method, and run the same tests again, this is what happens:

4 run, 1 failed, 0 errors
Failures:
  testRemove: <SetVector: <Vector: 2 3>> contains 3 after it was removed

debugging **method** testRemove ...

Whoops! We made a mistake when we implemented the *remove* method; it looks like it doesn’t actually remove the argument. The one line summary really tells us all that we need to know. The debugging information isn’t all that useful: it tells us that the assertion failed, and then gives a stack trace of the path through GUnit, which isn’t what we want. Perhaps one day Grace will have a debugger that will let us go back in time through the test that failed. Until then, the test itself should give you most of the information that you need to fix the problem. If I go back and correct my implementation of *remove*, this is the output on the next run:

4 run, 0 failed, 1 error
Errors:
  testRemove

debugging **method** testRemove ...

Error around line 231: RuntimeError: undefined value used as argument to []:=
Called Block«GUnit:134»._apply (defined at GUnit:128 in object at unknown:0) on line 140
Called MirrorMethod.request (defined at unknown:0 in object at unknown:0) on line 139
Called aSetTest.testRemove (defined at SetTests:36 in object at SetTests:44) on line 139
Called SetVector.remove (defined at set_vector:78 in object at set_vector:51) on line 37
Called VectorClass.removeValue (defined at vector:116 in object at vector:335) on line 79
Called VectorClass.removeFromIndex (defined at vector:222 in object at vector:335) on line 121
Called NativePrelude.while()do (defined at unknown:0 in object at unknown:0) on line 230
Called Block«vector:230»._apply (defined at unknown:0 in object at unknown:0) on line 230

```plaintext
230: while{i <= elementCount} do {
231:       elementData[ix] := elementData[ix + 1]
232:       ix := ix + 1
```

minigrace: Program exited with error: SetTests

This output is telling us about an error at line 231 of module vector, where the code accesses an undefined array element in method removeFromIndex. Perhaps we should now write some unit tests for the vector module.